AFFINE ISOMETRIES IN THE SPACE.

ROTATIONS

Let $E_3 = R^3$ be the euclidean affine plane, with the standard scalar product.

Unless otherwise stated we will work with the standard coordinate system $R = \{O, B = \{e_1, e_2, e_3\}\}$, which is an orthonormal system. *We* denote by (x, y, z) the coordinates of an arbitrary poin in E_3.

We will construct the matrix $M_f(R)$ of a rotation f:$E_3 \to E_3$ from the matrices of the next basic rotations of angle α :

Rotation Axis (line of fixed points)	y=0 z=0	x=0 z=0	x=0 y=0
$M_f(R)$	[[1, 0, 0, 0], [0, 1, 0, 0], [0, 0, cos(α), -sin(α)], [0, 0, sin(α), cos(α)]]	[[1, 0, 0, 0], [0, cos(α), 0, -sin(α)], [0, 0, 1, 0], [0, sin(α), 0, cos(α)]]	[[1, 0, 0, 0], [0, cos(α), -sin(α), 0], [0, sin(α), cos(α), 0], [0, 0, 0, 1]]

Given the vector w=(1,2,1) in E_3. Obtain a orthonormal basis {u1,u2} of the orthogonal subspace to w.

```
> restart:with(linalg):
> w:=[1,2,1]:
```

The orthogonal subspace U to w has equation:

```
> v:=[v1,v2,v3]:
> equ:=dotprod(v,w)=0;
```

$$equ := v1 + 2\,v2 + v3 = 0 \tag{1.1}$$

We obtain a basis of U is:

```
> solve(equ,{v1,v2,v3});
```

$$\{v1 = -2\,v2 - v3, v2 = v2, v3 = v3\} \tag{1.2}$$

```
> u:=[-2,1,0];[-1,0,1];
```

$$u := [-2, 1, 0]$$

$$[-1, 0, 1] \quad (1.3)$$

But it is not a orthogonal basis. We take u1 to be one of them (but unitary) and look for an orthogonal vector in U orthogonal to u1.

```
> u1:=(1/sqrt(dotprod(u,u)))*u;
```

$$u1 := \frac{1}{5}\sqrt{5}\,[-2, 1, 0] \quad (1.4)$$

```
> equ2:=dotprod(v,u1);
```

$$equ2 := -\frac{2}{5}\,v1\,\sqrt{5} + \frac{1}{5}\,v2\,\sqrt{5} \quad (1.5)$$

The vector u2 has to be orthogonal to w and to u1, and unitary.

```
> solve({equ,equ2},{v1,v2,v3});
```

$$\{v1 = v1, v2 = 2\,v1, v3 = -5\,v1\} \quad (1.6)$$

```
> t:=[1,2,-5];
```

$$t := [1, 2, -5] \quad (1.7)$$

```
> u2:=(1/sqrt(dotprod(t,t)))*t;
```

$$u2 := \frac{1}{30}\sqrt{30}\,[1, 2, -5] \quad (1.8)$$

```
>
```

We check that we did it right.

```
> dotprod(u1,w);dotprod(u2,w);dotprod(u1,u2);
```

$$0$$
$$0$$
$$0 \quad (1.9)$$

```
> dotprod(u1,u1);dotprod(u2,u2);
```

$$1$$
$$1 \quad (1.10)$$

In E_3, determine the matrix of the following movement: rotation of angle $\frac{\pi}{3}$ and axis the line r=(0,-1,3)+<(1,0,-5)>, compounded with translation of vector (2,0,1), with respect to the orthonormal coordinate system R={O, {e1,e2,e3}}.

```
> restart; with(linalg):
> T:=matrix([[1,0,0,0],[2,1,0,0],[0,0,1,0],[1,0,0,1]]);
```

(2.1)

$$T := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 2 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 1 \end{bmatrix} \quad (2.1)$$

```
> P:=matrix(3,1,[0,-1,3]); u[1]:=evalm((1/sqrt(1+25))*matrix(3,1,
  [1,0,-5]));
```

$$P := \begin{bmatrix} 0 \\ -1 \\ 3 \end{bmatrix}$$

$$u_1 := \begin{bmatrix} \frac{1}{26}\sqrt{26} \\ 0 \\ -\frac{5}{26}\sqrt{26} \end{bmatrix} \quad (2.2)$$

Matrix of the rotation in the orthonormal coordinate systeml R'={P,{u1,u2,u3}}

```
> alpha:=Pi/3;
```

$$\alpha := \frac{1}{3}\pi \quad (2.3)$$

```
> giro:=matrix([[1,0,0,0],[0,1,0,0],[0,0,cos(alpha), -sin(alpha)
  ],[0,0,sin(alpha), cos(alpha)]]);
```

$$giro := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \frac{1}{2} & -\frac{1}{2}\sqrt{3} \\ 0 & 0 & \frac{1}{2}\sqrt{3} & \frac{1}{2} \end{bmatrix} \quad (2.4)$$

We look for the vectors u2 and u3. They are a orthonormal basis of the orthogonal space to u1.

```
> v:=matrix(3,1,[v1,v2,v3]);
```

$$v := \begin{bmatrix} v1 \\ v2 \\ v3 \end{bmatrix} \quad (2.5)$$

```
> p:=evalm(transpose(u[1])&*v);ecu1:=p[1,1]=0;solve(ecu1);
```

$$p := \begin{bmatrix} \frac{1}{26}\sqrt{26}\, v1 - \frac{5}{26}\sqrt{26}\, v3 \end{bmatrix}$$

$$ecu1 := \frac{1}{26}\sqrt{26}\ v1 - \frac{5}{26}\sqrt{26}\ v3 = 0$$

$$\{v1 = 5\ v3, v3 = v3\} \quad (2.6)$$

```
> u[2]:=evalm((1/sqrt(1+25))*matrix(3,1,[5,0,1]));
```

$$u_2 := \begin{bmatrix} \frac{5}{26}\sqrt{26} \\ 0 \\ \frac{1}{26}\sqrt{26} \end{bmatrix} \quad (2.7)$$

```
> ecu2:=evalm(transpose(u[2])&*v)[1,1]=0; solve({ecu1,ecu2});
```

$$ecu2 := \frac{5}{26}\sqrt{26}\ v1 + \frac{1}{26}\sqrt{26}\ v3 = 0$$

$$\{v1 = 0, v3 = 0\} \quad (2.8)$$

```
> u[3]:=matrix(3,1,[0,1,0]);
```

$$u_3 := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix} \quad (2.9)$$

```
> print(`Matrices of the change of coordiantes`); RpR:=
  stackmatrix(matrix(1,4,[1,0,0,0]),concat(P,u[1],u[2],u[3]));
  RRp:=inverse(RpR);
```

Matrices of the change of coordiantes

$$RpR := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & \frac{1}{26}\sqrt{26} & \frac{5}{26}\sqrt{26} & 0 \\ -1 & 0 & 0 & 1 \\ 3 & -\frac{5}{26}\sqrt{26} & \frac{1}{26}\sqrt{26} & 0 \end{bmatrix}$$

$$RRp := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{15}{26}\sqrt{26} & \frac{1}{26}\sqrt{26} & 0 & -\frac{5}{26}\sqrt{26} \\ -\frac{3}{26}\sqrt{26} & \frac{5}{26}\sqrt{26} & 0 & \frac{1}{26}\sqrt{26} \\ 1 & 0 & 1 & 0 \end{bmatrix} \quad (2.10)$$

The matrix of the rotation in the coordinate system R is.

```
> G:=evalm(RpR&*giro&*RRp);
```

$$G := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{15}{52} - \frac{5}{52}\sqrt{26}\sqrt{3} & \frac{27}{52} & -\frac{5}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} \\ -\frac{1}{2} - \frac{3}{52}\sqrt{26}\sqrt{3} & \frac{5}{52}\sqrt{26}\sqrt{3} & \frac{1}{2} & \frac{1}{52}\sqrt{26}\sqrt{3} \\ \frac{3}{52} - \frac{1}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} & -\frac{1}{52}\sqrt{26}\sqrt{3} & \frac{51}{52} \end{bmatrix} \quad \textbf{(2.11)}$$

The matrix of the composition of the rotation witht the translation in the coordinate system R is

```
> TG:=simplify(evalm(T&*G));
```

$$TG := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{119}{52} - \frac{5}{52}\sqrt{26}\sqrt{3} & \frac{27}{52} & -\frac{5}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} \\ -\frac{1}{2} - \frac{3}{52}\sqrt{26}\sqrt{3} & \frac{5}{52}\sqrt{26}\sqrt{3} & \frac{1}{2} & \frac{1}{52}\sqrt{26}\sqrt{3} \\ \frac{55}{52} - \frac{1}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} & -\frac{1}{52}\sqrt{26}\sqrt{3} & \frac{51}{52} \end{bmatrix} \quad \textbf{(2.12)}$$

Let us suppose that we are given the matrix TG and we have to classify this isometry.

```
> A:=submatrix(TG,2..4,2..4); AI:=evalm(A-diag(1,1,1)); b:=
  submatrix(TG,2..4,1..1);
```

$$A := \begin{bmatrix} \frac{27}{52} & -\frac{5}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} \\ \frac{5}{52}\sqrt{26}\sqrt{3} & \frac{1}{2} & \frac{1}{52}\sqrt{26}\sqrt{3} \\ -\frac{5}{52} & -\frac{1}{52}\sqrt{26}\sqrt{3} & \frac{51}{52} \end{bmatrix}$$

$$AI := \begin{bmatrix} -\frac{25}{52} & -\frac{5}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} \\ \frac{5}{52}\sqrt{26}\sqrt{3} & -\frac{1}{2} & \frac{1}{52}\sqrt{26}\sqrt{3} \\ -\frac{5}{52} & -\frac{1}{52}\sqrt{26}\sqrt{3} & -\frac{1}{52} \end{bmatrix}$$

(2.1.1)

$$b := \begin{bmatrix} \frac{119}{52} - \frac{5}{52}\sqrt{26}\sqrt{3} \\ -\frac{1}{2} - \frac{3}{52}\sqrt{26}\sqrt{3} \\ \frac{55}{52} - \frac{1}{52}\sqrt{26}\sqrt{3} \end{bmatrix} \qquad (2.1.1)$$

```
> rank(AI); rank(concat(b,AI));
```

$$2$$

Warning, unable to find a provably non-zero pivot

$$3 \qquad (2.1.2)$$

Decompose the obtained isometry as the rotation of axis the invariant line compounded with a translation of vector parallel to the rotation axis.

Rotation angle:

```
> solve({trace(A)-2*cos(theta)-1},{theta});
```

$$\left\{\theta = \frac{1}{3}\pi\right\} \qquad (2.2.1)$$

The rotation axis is an invariant line in the direction of the eigenvector associated to the eigenvalue 1.

```
> eigenvectors(A);
```

$$\left[\frac{1}{2} + \frac{1}{2} I\sqrt{3}, 1, \left\{\left[\begin{array}{ccc} 5 & -\frac{2}{3}\sqrt{26}\sqrt{3}\left(\frac{1}{2} + \frac{1}{2} I\sqrt{3}\right) + \frac{1}{3}\sqrt{26}\sqrt{3} & 1 \end{array}\right]\right\}\right], \left[\frac{1}{2} - \frac{1}{2} I\sqrt{3}, 1, \left\{\left[\begin{array}{ccc} 5 & -\frac{2}{3}\sqrt{26}\sqrt{3}\left(\frac{1}{2} - \frac{1}{2} I\sqrt{3}\right) + \frac{1}{3}\sqrt{26}\sqrt{3} & 1 \end{array}\right]\right\}\right], \left[1, 1, \left\{\left[\begin{array}{ccc} -\frac{1}{5} & 0 & 1 \end{array}\right]\right\}\right] \qquad (2.2.2)$$

So we can take u1 as before.

```
> u[1]:=evalm((1/sqrt(1+25))*matrix(3,1,[1,0,-5]));
```

$$u_1 := \begin{bmatrix} \frac{1}{26}\sqrt{26} \\ 0 \\ -\frac{5}{26}\sqrt{26} \end{bmatrix} \qquad (2.2.3)$$

```
> X:=matrix(4,1,[1,x[1],x[2],x[3]]);
```

(2.2.4)

$$X := \begin{bmatrix} 1 \\ x_1 \\ x_2 \\ x_3 \end{bmatrix} \tag{2.2.4}$$

```
> Xf(X):=simplify(evalm(TG&*X-X));
```

$$Xf(X) := \begin{bmatrix} 0 \\ \frac{119}{52} - \frac{5}{52}\sqrt{26}\sqrt{3} - \frac{25}{52}x_1 - \frac{5}{52}\sqrt{26}\sqrt{3}\,x_2 - \frac{5}{52}x_3 \\ -\frac{1}{2} - \frac{3}{52}\sqrt{26}\sqrt{3} + \frac{5}{52}\sqrt{26}\sqrt{3}\,x_1 - \frac{1}{2}x_2 + \frac{1}{52}\sqrt{26}\sqrt{3}\,x_3 \\ \frac{55}{52} - \frac{1}{52}\sqrt{26}\sqrt{3} - \frac{5}{52}x_1 - \frac{1}{52}\sqrt{26}\sqrt{3}\,x_2 - \frac{1}{52}x_3 \end{bmatrix} \tag{2.2.5}$$

```
> ecua1:=Xf(X)[2,1]-a*u[1][1,1]; ecua2:=Xf(X)[3,1]-a*u[1][2,1];
  ecua3:=Xf(X)[4,1]-a*u[1][3,1];
```

$$ecua1 := \frac{119}{52} - \frac{5}{52}\sqrt{26}\sqrt{3} - \frac{25}{52}x_1 - \frac{5}{52}\sqrt{26}\sqrt{3}\,x_2 - \frac{5}{52}x_3 - \frac{1}{26}a\sqrt{26}$$

$$ecua2 := -\frac{1}{2} - \frac{3}{52}\sqrt{26}\sqrt{3} + \frac{5}{52}\sqrt{26}\sqrt{3}\,x_1 - \frac{1}{2}x_2 + \frac{1}{52}\sqrt{26}\sqrt{3}\,x_3$$

$$ecua3 := \frac{55}{52} - \frac{1}{52}\sqrt{26}\sqrt{3} - \frac{5}{52}x_1 - \frac{1}{52}\sqrt{26}\sqrt{3}\,x_2 - \frac{1}{52}x_3 + \frac{5}{26}a\sqrt{26} \tag{2.2.6}$$

```
> sol:=solve({ecua1,ecua2,ecua3},{x[1],x[2],a});
```

$$sol := \left\{a = -\frac{3}{26}\sqrt{26},\ x_1 = \frac{17}{10} - \frac{1}{5}x_3,\ x_2 = -1 + \frac{11}{52}\sqrt{26}\sqrt{3}\right\} \tag{2.2.7}$$

The invariant line has equations `x[1] = 17/10-1/5*x[3]; x[2] = -1+11/52*26^(1/2)*3^(1/2);`

$$x_1 = \frac{17}{10} - \frac{1}{5}x_3$$

$$x_2 = -1 + \frac{11}{52}\sqrt{26}\sqrt{3} \tag{2.2.8}$$

To obtain a translation vector parallel to this line we take a point Q in the line and compute u=Qf(Q)

```
> assign(sol):x[3]:=-5;
```

$$x_3 := -5 \tag{2.2.9}$$

```
> Q:=matrix(4,1,[1,x[1],x[2],x[3]]);
```

(2.2.10)

$$Q := \begin{bmatrix} 1 \\ \frac{27}{10} \\ -1 + \frac{11}{52}\sqrt{26}\sqrt{3} \\ -5 \end{bmatrix} \quad (2.2.10)$$

The invariant line is Q+<u> were.

```
> u:=simplify(evalm(TG&*Q-Q));
```

$$u := \begin{bmatrix} 0 \\ -\frac{3}{26} \\ 0 \\ \frac{15}{26} \end{bmatrix} \quad (2.2.11)$$

```
> Tu:=matrix(4,4,[1,0,0,0,-3/26,1,0,0,0,0,1,0,15/26,0,0,1]);
```

$$Tu := \begin{bmatrix} 1 & 0 & 0 & 0 \\ -\frac{3}{26} & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{15}{26} & 0 & 0 & 1 \end{bmatrix} \quad (2.2.12)$$

Then TG=Tu Ginv, where Ginv is the matrix of the rotation with axis the invariant line.

```
> Ginv:=evalm(inverse(Tu)&*TG);
```

$$Ginv := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{125}{52} - \frac{5}{52}\sqrt{26}\sqrt{3} & \frac{27}{52} & -\frac{5}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} \\ -\frac{1}{2} - \frac{3}{52}\sqrt{26}\sqrt{3} & \frac{5}{52}\sqrt{26}\sqrt{3} & \frac{1}{2} & \frac{1}{52}\sqrt{26}\sqrt{3} \\ \frac{25}{52} - \frac{1}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} & -\frac{1}{52}\sqrt{26}\sqrt{3} & \frac{51}{52} \end{bmatrix} \quad (2.2.13)$$

TG and Ginv have the same matrix in the coordinate system R!

```
> evalm(TG);
```

(2.2.14)

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{119}{52} - \frac{5}{52}\sqrt{26}\sqrt{3} & \frac{27}{52} & -\frac{5}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} \\ -\frac{1}{2} - \frac{3}{52}\sqrt{26}\sqrt{3} & \frac{5}{52}\sqrt{26}\sqrt{3} & \frac{1}{2} & \frac{1}{52}\sqrt{26}\sqrt{3} \\ \frac{55}{52} - \frac{1}{52}\sqrt{26}\sqrt{3} & -\frac{5}{52} & -\frac{1}{52}\sqrt{26}\sqrt{3} & \frac{51}{52} \end{bmatrix} \quad \textbf{(2.2.14)}$$